




# Opencockpits



## Manual IOCard USB Outputs

## Índice:

MANUAL IOCARD USB OUTPUTS.....	1
ÍNDICE:.....	2
INTRODUCCIÓN:.....	3
USB OUTPUTS: .....	3
ESQUEMA Y COMPONENTES:.....	3
DESCRIPCIÓN DE LOS CONECTORES: .....	4
ESQUEMA DE CONEXIÓN:.....	5
PUESTA EN MARCHA DE LA TARJETA: .....	6
TEST_OUTPUTS: .....	7
SIOC MONITOR: .....	7
INSTALACIÓN Y CONFIGURACIÓN: .....	9
DECLARACIÓN EN SIOC.INI: .....	9
<i>Salidas:</i> .....	9
<i>Entradas analógicas:</i> .....	9
DECLARACIÓN EN SCRIPTS: .....	9
<i>Entradas analógicas:</i> .....	9
<i>Salidas:</i> .....	10
<i>Control de Intensidad de salidas:</i> .....	10
EJEMPLOS DE SCRIPTS:.....	10
<i>Salidas:</i> .....	10
<i>Control de intensidad de salidas:</i> .....	11
<i>Entradas analógicas:</i> .....	13
LINKS DE INTERÉS:.....	13


## Introducción:

La tarjeta IOCard USB Outputs ha sido diseñada para poder manejar 64 salidas y gestionar el control de intensidad de las mismas y además posee 5 entradas analógicas. La conexión de la tarjeta al ordenador es mediante el puerto USB.

### Usb Outputs:


La tarjeta se autoinstala como dispositivo HID y se controla mediante el protocolo IOCP con SIOC.

Cada una de las 64 salidas puede tener una carga de hasta 50V y 500mA con un máximo de 2,5A para la tarjeta entera. El control de intensidad es de 7 bits (0-127) mediante PWM (Pulse Width Modulation) que puede usarse para controlar leds o lámparas para indicadores, así como de la gestión integral de las retroiluminaciones de paneles, eso si, la intensidad siempre será la misma para todas las salidas, no puede gestionarse independientemente para cada salida.



### Esquema y componentes:

- C1 = CONDENSADOR 220nf
- C2, C3 = CONDENSADORES 22pF
- C4 A C14 = CONDENSADORES 0.1mF
- D1 = DIODOS LED
- IC1 = MICROCONTROLADOR 16C745
- IC2 = CIRCUITO INTEGRADO 74HC154
- IC3, IC10 = CIRCUITO INTEGRADO 74HC259
- IC11, IC18 = CIRCUITO INTEGRADO ULN2803
- J1 = CONECTOR USB
- J2 = CONECTOR ALIMENTACION +5V (no protegida contra inversión de polaridad)
- J3 = CONECTOR ALIMENTACION EXTERNA (no protegida contra inversión de polaridad)
- JMP1 = JUMPER SELECTOR ALIMENTACION EXTERNA/INTERNA
- JMP2 = JUMPER SELECTOR ALIMENTACION EXTERNA/INTERNA
- J4 (AN-0), J5 (AN-1), J6 (AN-2), J7 (AN-3), J8 (AN-4) = CONECTORES ENTRADAS ANALÓGICAS DE 3 PINES
- Q1 = CRISTAL CUARZO 6Mhz
- R1 = RESISTENCIA 1K5
- R2 = RESISTENCIA 10K
- R3 = RESISTENCIA 470R
- T1 A T22 = CONECTORES A PCB DE TORNILLO DE 3 PINES



**Descripción de los conectores:**


- J1 = Conector USB al ordenador. El ordenador reconocerá la tarjeta automáticamente como un dispositivo HID.
- T2 = Conectores de tornillo para las salidas, numerados del 0 al 63. Importante: En estos conectores enchufaremos los negativos (cátodos, k o GND) de los leds, lámparas u otros elementos.
- POWER = Conector para la alimentación de los elementos conectados a las salidas, La tensión de alimentación estará entre 3V y 40V no más (la placa no está protegida contra inversión de polaridad).
- J2 = Conector de entrada para alimentación de la IOcard USB Outputs desde fuera, +5V (la placa no está protegida contra inversión de polaridad).
- J3 = Conector de salida de alimentación con la misma tensión que en los conectores de tornillo T procedente de POWER.
- J4 a J8 = Conectores para las entradas analógicas: AN-0 a AN-4.
- JMP1 y JMP2 = Puentes de configuración para la alimentación de la tarjeta, según tabla adjunta:

JUMPER 1	JUMPER 2	ALIM. PLACA	ALIM. SALIDAS
ABIERTO	ABIERTO	EXTERNA	EXTERNA
CERRADO	ABIERTO	USB	EXTERNA
CERRADO	CERRADO	USB	EXTERNA (5V)
ABIERTO	CERRADO	-	-

**Atención:** Si va conectar una fuente de alimentación para las salidas de un valor diferente a +5V y mantiene cerrados los Jumpers 1 y 2 la tarjeta se dañará.

**Esquema de conexión:**

A continuación se muestra un ejemplo de conexionado de varios elementos a la tarjeta, así como las conexiones de alimentación, tanto de entrada como de salida:




Tenemos que tener en cuenta que el común de todos los elementos conectados es el positivo (ánodo, A), para las lámparas no importa la polaridad pero para los leds si.



## Puesta en marcha de la tarjeta:


Ya conocemos la teoría de conexión de la IOCard USB Outputs, ahora pasemos a comprobarla y conectarla para ver los resultados.

Al final de este documento hay un listado de links para poder descargar el software necesario para poder poner en práctica este manual.



Ha sido reconocida, a partir de aquí tenemos dos opciones para probar su funcionamiento, una con test\_outputs.exe, programa hecho exclusivamente para testear esta tarjeta con todas sus opciones incluida la de nivel de intensidad de la salida y la otra con SIOC Monitor, que no controla el nivel de intensidad de la salida.


Pero para poder probarla tendremos que montar algún circuito de ejemplo:



En el ejemplo hemos conectado unos led rojos a la salidas 33 y 56 y uno blanco en la 43, además de un potenciómetro en AN0 (J4) y puentes en el resto de entradas analógicas para que no fluctúen sus lecturas. La alimentación escogida para la placa ha sido la USB y la de las salidas externa de sólo 5V (JMP1 y JMP2 cerrados)

## Test\_outputs:

Este programa se ejecuta directamente y muestra una imagen de la placa con todas sus entradas y salidas, a las que tenemos acceso con el ratón para activar o desactivar, además de la lectura de las entradas analógicas y el control deslizante para cambiar la intensidad de las salidas (todas tendrán la misma intensidad siempre).




Advertencia: en cuanto conectemos el USB el led rojo D1 permanecerá encendido a menos que haya algún problema.

Podemos hacer pruebas ahora de si nuestro montaje funciona pulsando ALL ON y después CLEAR o dando a salida por salida para activarla o desactivarla, además de poder jugar con la intensidad de la salida (los leds cambiarán de intensidad) y la lectura del potenciómetro conectado.

## Sioc Monitor:

Sioc tiene integrado un sistema de comprobación para todas las IOCards, para la USB Outputs también pero sin el control de intensidad de las salidas.




Ahora probaremos con SIOC Monitor si nuestro montaje ha salido bien y todo funciona. Arrancamos SIOC, miramos si nos la ha reconocido y si es así pulsamos el botón Monitor, en IOCard USB Outputs y chequeamos la instalación:

The screenshot shows the SIOC Monitor software interface for the IOCard USB Outputs device. The interface is divided into several sections:

- DIGITAL INPUTS:** A grid of 288 pins (0-287) arranged in 8 rows and 36 columns.
- DIGITAL OUTPUTS:** A grid of 288 pins (0-287) arranged in 8 rows and 36 columns.
- ANALOG INPUTS:** 5 channels (#1 to #5). Channel #1 shows a value of 191. A "Not Show" checkbox is present.
- DISPLAY DIGITS:** Controls for "Position" and "Value 0-15". Both are set to 0. "Initial Pos." and "Number" are also set to 0. "Send" buttons are available for each.
- COMMANDS:** "ALL ON" and "ALL OFF" buttons.
- Legend:** Not available (grey), Reserved (red), Status ON (green), Status OFF (white).
- EXIT** button is located at the bottom right.

Como podemos observar la entrada analógica #1 muestra el valor de nuestro potenciómetro y las demás muestran el valor 0, si giramos el potenciómetro veremos cómo cambia el valor representado de más a menos o al revés (depende de cómo hagamos las conexiones de los pines), si no nos interesa el sentido de subida o bajada de los valores podemos intercambiar los pines exteriores del conector J4 o hacerlo por software mediante un script SIOC.

Probemos ahora las salidas, si pulsamos el botón ALL ON se activarán todas las salidas y nuestros 3 leds.





Deberíamos hacer estas pruebas para todos los conectores.

## Instalación y configuración:

Para poder hacer uso de las características de la USB Outputs debemos tener instalado SIOC (última versión a ser posible) y el simulador de vuelo FS, FSX, Xplane, etc.

Para gestionar esta tarjeta se usan scripts, para controlar la tarjeta deberemos editar la entrada en el fichero sioc.ini, de tal manera que asignemos un índice de dispositivo a cada tarjeta que instalemos, creando una entrada en el fichero sioc.ini por cada tarjeta USB Outputs conectada. Es importante recordar que la USB Outputs tiene dos funciones, una la de control de salidas con gestión de la intensidad de las mismas y la otra de entradas analógicas que también deben ser declaradas en sioc.ini, en el siguiente formato:

### **Declaración en SIOC.ini:**

#### **Salidas:**

**MASTER=XX,6,1,YY**

**XX** indica el número de índice (IDX) dentro de nuestro sistema de tarjetas.

**YY** indica el número de dispositivo del puerto USB al que está conectado (DEVICE, 122 en nuestro ejemplo).

#### **Entradas analógicas:**

**USBAnalogic=XX,YY**

**XX** indica el número de índice (IDX) dentro de nuestro sistema de tarjetas.

**YY** indica el número de dispositivo del puerto USB al que está conectado (DEVICE).

Ambos deben ser los mismos que en la declaración de las salidas.

Por ejemplo, si conectamos dos tarjetas USB Outputs con los números de dispositivos 94 y 122 (estos números se pueden averiguar fácilmente con el propio programa SIOC) entonces las declararíamos en el sioc.ini de la siguiente manera:

MASTER=1,6,1,122 [como en nuestro ordenador de pruebas]

MASTER=2,6,1,94

USBAnalogic=1,122 [como en nuestro ordenador de pruebas]

USBAnalogic=2,94

### **Declaración en scripts:**

#### **Entradas analógicas:**

Para la lectura de las entradas analógicas se deberá usar el formato siguiente:

**Var VVVV, name NNNN, Link IOCARD\_ANALOGIC, Input EE, posL LLL, posC CCC, posR RRR, Device DD**

**VVVV** = número variable.

**NNNN** = nombre variable (opcional).

**EE** = numero de entrada analógica 1-5.

**LLL** = posición máxima del dispositivo a la izquierda.

**CCC** = posición central del dispositivo.

**RRR** = posición máxima del dispositivo a la derecha.

**DD** = numero de índice definido en sioc.ini (si la tenemos declarada como 0: no hace falta poner la referencia al número de Device).

Ejemplo de definición de entrada analógica:

```
Var 0002, name pot_flaps, Link IOCARD_ANALOGIC, Input 2, posL 1, posC 128, posR 255, Device 1
```

## Salidas:

Para hacer referencia al número de salida exacta, debemos tener en cuenta el número de índice que le hemos asignado a cada una de las tarjetas USB Outputs. Debemos definir la salida:

**Var VVVV, name NNNN, Link IOCARD\_OUT, Output SS, Device DD**

**VVVV** = número variable.

**NNNN** = nombre variable (opcional).

**SS** = numero de salida 0-63.

**DD** = numero de índice definido en sioc.ini (si la tenemos declarada como 0: no hace falta poner la referencia al número de Device).

Ejemplo de definición:

```
Var 0001, name bright_ctrl, Link IOCARD_OUT, Output 33, Device 1
```

## Control de Intensidad de salidas:

Para controlar la intensidad de las salidas se definirán usando el formato de un display (Link IOCARD\_DISPLAY) pero asignando un valor fijo al dígito (Digit 1) y al número (Numbers 3). Esta variable puede tomar valores entre 0 (mínima intensidad) y 127 (máxima intensidad):

**Var VVVV, name NNNN, Link IOCARD\_DISPLAY, DEVICE XX, Digit 1, Numbers 3**

**VVVV** = número variable.

**NNNN** = nombre variable (opcional).

**XX** = device declarado en sioc.ini.

Ejemplo de definición:

```
Var 0001, Name bright, Link IOCARD_DISPLAY, Device 1, Digit 1, Numbers 3
```

## Ejemplos de scripts:

### Salidas:

El ejemplo que hemos usado anteriormente nos servirá para probar el funcionamiento de la USB Outputs, nos basaremos en el montaje de prueba del capítulo Puesta en Marcha de la tarjeta, con la finalidad de tener 3 testigos para el estado de la rueda delantera del tren de aterrizaje: Led rojo de la salida 56 encendido sólo para tren totalmente recogido, led rojo de la salida 33 encendido sólo para tren totalmente fuera y led blanco de la salida 43 encendido sólo para tren en tránsito.

Para empezar crearemos un archivo de texto llamado "prueba\_usb\_outputs1.txt":

```
// *****
// * Config_SIOC ver 4.01 - By Manolo Vélez - www.opencockpits.com
// *****
// * FileName : prueba_usb_outputs1.txt
// * Date : 10/02/2012
Var 0001, Name ledarriba, Link IOCARD_OUT, Output 56, Device 1 // Indicador de tren recogido
Var 0002, Name ledtransito, Link IOCARD_OUT, Output 43, Device 1 // Indicador de tren en tránsito
Var 0003, Name ledabajo, Link IOCARD_OUT, Output 33, Device 1 // Indicador de tren fuera
Var 0004, Name tren, Link FSUIPC_INOUT, Offset $0BEC, Length 4 // Offset que lee la posición del tren
delantero
{
IF &tren = 0 // SI el tren está recogido
{
&ledarriba = 1 // el indicador de tren arriba está encendido
&ledtransito = 0 // el indicador de tren en tránsito está apagado
&ledabajo = 0 // el indicador de tren abajo está apagado
}
IF &tren >= 1 // SI el tren está bajando o subiendo
{
IF &tren < 16383 // y SI además el tren no está abajo del todo
{
&ledarriba = 0 // el indicador de tren arriba está apagado
&ledtransito = 1 // el indicador de tren en tránsito está encendido
&ledabajo = 0 // el indicador de tren abajo está apagado
}
}
IF &tren = 16383 // SI el tren está abajo totalmente
{
&ledarriba = 0 // el indicador de tren arriba está apagado
&ledtransito = 0 // el indicador de tren en tránsito está apagado
&ledabajo = 1 // el indicador de tren abajo está encendido
}
}
//End of file prueba_usb_outputs1.txt
```

Escribiremos el script lo guardaremos y lo ejecutaremos con SIOC, arrancamos FS y observamos que el led rojo asignado a tren abajo está encendido (salida 33) y los otros dos apagados. Despegamos y una vez arriba, subimos el tren observaremos que se apaga el led de tren fuera, se enciende el led de tránsito (el blanco de la salida 43) y al momento se apaga de nuevo y se enciende el led asignado a tren recogido (el rojo de la salida 56) y si sacamos el tren el proceso será a la inversa.

## Control de intensidad de salidas:

En este caso aprenderemos a cambiar la intensidad de las salidas por software y por hardware.

### Hardware.

Para este ejemplo nos hará falta el simulador, con el montaje que tenemos de la prueba anterior nos basta y haremos uso del potenciómetro conectado a la entrada AN-0 o J4. En este caso lo que queremos ver es como cambia la intensidad de luz de las salidas, haciendo que el led blanco que está conectado a la salida 43 se encienda cuando el tren esté abajo además de poder controlar su intensidad de luz con el potenciómetro, para ello escribiremos un nuevo script:

```
// *****
// * Config_SIOC ver 4.01 - By Manuel Velez - www.opencockpits.com
// *****
// * FileName : prueba_intensidad_hardware_usb_outputs.txt
// * Date : 16/02/2012
Var 0001, name ledabajo, Link IOCARD_OUT, Device 1, Output 43 // Led en salida 43 encendido
Var 0002, name dimmer, Link IOCARD_ANALOGIC, Device 1, Input 1, PosL 1, PosC 128, PosR 255
// Lectura del potenciómetro
{
  &bright = &dimmer
}
Var 0003, name bright, Link IOCARD_DISPLAY, Device 1, Digit 1, Numbers 3 // dimmer at USB Outputs
Var 0004, name tren, Link FSUIPC_INOUT, Offset $0BEC, Length 4 // Offset que lee la posición del tren
{
  IF &tren <= 16383
  {
    &ledabajo = 1
  }
}
// End of file prueba_intensidad_hardware_usb_outputs.txt
```

Escribiremos el script, lo guardaremos y lo ejecutaremos con SIOC, arrancamos FS, seleccionamos un avión con tren retráctil y elegimos salir desde el aeropuerto, observaremos que el led blanco asignado a tren abajo está encendido (salida 43) y si movemos el potenciómetro debe cambiar su intensidad lumínica, si no hay que reparar la instalación.

## Software.

Para este ejemplo haremos lo mismo que en el anterior pero haciendo que varíe la intensidad de luz en forma de fluctuación de menos a más y al contrario sin necesidad de potenciómetro, para ello escribiremos un nuevo script:

```
// *****
// * Config_SIOC ver 4.01 - By Manuel Velez - www.opencockpits.com
// *****
// * FileName : prueba_intensidad_software_usb_outputs.txt
// * Date : 17/02/2012

Var 0001, name ledabajo, Link IOCARD_OUT, Device 1, Output 43 // Led de la salida 43 encendido
Var 0003, name bright, Link IOCARD_DISPLAY, Device 1, Digit 1, Numbers 3 // dimmer at USB Outputs
{
  IF &ledabajo = 1 // si el led de aviso está activo
  {
    &bright = DELAY 127 ,80 // espera 80 ms a subir la intensidad del led a 127
    &bright = DELAY 50 ,20 // espera 20 ms a bajar la intensidad del led a 50
  }
}
Var 0004, name tren, Link FSUIPC_INOUT, Offset $0BEC, Length 4 // Offset lee posición del tren
{
  IF &tren = 16383 // si el tren esta abajo
  {
    &ledabajo = 1 // led de aviso encendido
    &bright = 127 // led con toda la intensidad
  }
}
// End of file
```

Guardamos el archivo y lo ejecutamos con SIOC, arrancamos FS, seleccionamos un avión con tren retráctil y observaremos que el led blanco asignado a tren abajo está encendido (salida 43) y cambia de intensidad lumínica, si no cambia hay que reparar la instalación. (Este ejemplo no tiene en cuenta ningún cambio posterior en el estado del tren).

## Entradas analógicas:

Para las entradas analógicas que tiene la USB Outputs usaremos el mismo ejemplo que para el control de la intensidad de luz por hardware ya que usa un potenciómetro conectado a la propia tarjeta. Estas entradas analógicas pueden ser usadas para otros fines distintos al del control de intensidad de la USB Outputs.

Recordemos que lo que queremos ver es como se usa una entrada analógica, cambiando la intensidad de luz de las salidas, haciendo que el led blanco que está conectado a la salida 43 cambie su brillo, para ello usaremos de nuevo el script:

```
// *****
// * Config_SIOC ver 4.01 - By Manuel Velez - www.opencockpits.com
// *****
// * FileName : prueba_entradas_analogicas_usb_outputs.txt
// * Date : 16/02/2012
Var 0001, name ledabajo, Link IOCARD_OUT, Device 1, Output 43 // Led en salida 43 encendido
Var 0002, name dimmer, Link IOCARD_ANALOGIC, Device 1, Input 1, PosL 1, PosC 128, PosR 255
// Lectura del potenciómetro
{
  &bright = &dimmer
}
Var 0003, name bright, Link IOCARD_DISPLAY, Device 1, Digit 1, Numbers 3 // dimmer at USB Outputs
Var 0004, name tren, Link FSUIPC_INOUT, Offset $0BEC, Length 4 // Offset que lee la posición del tren
{
  IF &tren <= 16383
  {
    &ledabajo = 1
  }
}
// End of file prueba_entradas_analogicas_usb_outputs.txt
```

Escribiremos el script, lo guardaremos y lo ejecutaremos con SIOC, arrancamos FS, seleccionamos un avión con tren retráctil y elegimos salir desde el aeropuerto, observaremos que el led blanco asignado a tren abajo está encendido (salida 43) y cambiará sólo el brillo, si no, hay que repasar la instalación.

Con esto damos fin a este manual, os invitamos a leer los manuales de los demás elementos de Opencockpits y del software SIOC y os damos las gracias por confiar en nosotros.

## Links de interés:

Zona de soporte para clientes:

<http://www.opencockpits.com/catalog/info/>